
Presentamosun control deaccesode muy bajo costorealizadocon tarjetaso tagsRFID. Utilizamoscomo
lector almóduloGP8F-R2,demuy bajocosto,el queconectamosa un micro de la serie HT48E deHoltek para
validar los RFID. Dada la capacidad de memoriaEEPROMde estosmicros, podemosalmacenarhasta24
RFIDs en memoria con el más pequeño.

Descripción del GP8F-R2

El móduloGP8F-R2lee tarjetaso tagsRFID read-only de 64-bits,códigoManchestera 125KHz.Poseeuna
salidaparaconectarun LED, que permanece encendidoy aumentasuintensidadal aproximarun RFID. El ID
correspondiente seentregapor un terminalen formato 8 bits serie asincrónico, a 9600 bps, sinparidad,a nivel
lógico. El usuariopuedeoptar por conectarlo aalgun driver RS-232para leerlo desdeuna PC, o conectarlo
directamente a un microcontrolador o UART. El formato lógico responde al siguiente esquema, en ASCII:

<STX> <DATA (10 bytes)> <CR> <LF> <ETX>

STX: ASCII Start of Text (02h)
ETX: ASCII End of Text (03h)
CR: ASCII Carriage Return (0Dh)
LF: ASCII Line Feed (0Ah)

El campo de DATA esunarepresentaciónen ASCII del ID delRFID, representando5 bytesbinarios (40 bits)
de la siguiente forma:
byte ASCII
C1 43 31
Por ejemplo, el ID 60 22 57 C0 31  se transmite de la siguiente forma:

02 36 30 32 32 35 37 43 30 33 31 0D 0A 03

  Recordemos que este tipo de RFID tags de 64 bits 125 Khz utiliza 9 bits para header, 40 para datos, 14 bits de
paridad y uno de stop.

Descripción del sistema de acceso propuesto

Setratadeun simple control deaccesodonde al acercarun tag autorizado el sistema responde activandoun
terminal duranteun tiempo determinado.El usuariopuedeutilizar esteterminal paracontrolarun triac, un
relé, o lo que prefiera, que a su vez controlarán un solenoide o pasarán información a otro sistema.

La operatoria del sistema es bastante básica, dado que setrata de una nota de aplicación, sin embargo se
proveesoporteparaincorporaro autorizarnuevosIDs, medianteotro tag previamenteautorizado,y unapurga
o borrado total de la base.
  La operatoria propuesta es la siguiente:
1.  Al arrancar el sistema, verifica la existencia de IDs en memoria.
2. Si no existen IDs precargados, activamomentáneamentela salidaindicandoesta situación,y permanece a

la espera deun ID. Al ingresarseel primer ID, estepasa a ser el “maestro”, es decir, no seutilizará para
ingresar sino para autorizar otros IDs.

3. Si ya existen IDs precargados (al menos uno), el sistema permanece en espera deun ID que conozca,
respondiendoantecualquierade ellos diferente del “maestro”, pulsando elterminal de activación porun
tiempo prefijado.

4. Si el ID reconocidoes el “maestro”,y el pin de purgase encuentraen estadológico 0, seindica esta
condición mediantedos pulsaciones de lasalida. El sistema queda entonces a la espera deun ID
desconocido a autorizar, el cual memorizará, activando la salida. Si se acerca un tag con un ID conocido, se
indicará su rechazo mediante dos pulsaciones de la salida y quedando nuevamente a la espera. Puede salirse
de este estado acercando nuevamente la tarjeta “maestra”.

5. Si el ID reconocidoes el “maestro”,y el pin depurgaseencuentraenestadológico 1, seprocedea borrar
toda labasede datos,exceptola tarjeta“maestra”.Seindicaesta condiciónmediantetres pulsaciones de la
salida.

Los puntos másflojos de este sistema, enlos que el usuario deberádesarrollarun poco más, sonlos
siguientes:

Si se extravíaun ID autorizado,no esposibleeliminarlo de formaindividual, debiendopurgarsela basede
IDs, lo que ocasiona lapérdidade todoslos demás,exceptoel “maestro”.Implementarun borradoselectivo
requiereun grantrabajosobrela basede datosy algunainterfazde edición, escapando al alcance de esta nota
de aplicación.

La tarjetao ID “maestra”deberáguardarseen lugarseguro, dado que nopuedeperderseni caer en manos de
personas noautorizadas.De extraviarse,cosapocoprobablesi setomanlas debidasprecauciones, se deberá
recargarel firmware. Esto podríaresolversefácilmentemedianteun pin extra,que al detectarseactivo borre
toda labase,incluyendoel ID “maestro”. Sin embargo, preferimos no complicaraúnmás el desarrolloy dejar
pines libresparaconectar resonadoreso cristales (encasode tenerproblemas con el osciladorinterno),y una
posiblecomunicación con otro sistemao control de otrasalida.Queda a criterio del usuariofinal la última
palabra.

Hardware

Conectamosdirectamentela salidadel móduloGP8F-R2a un pin del microcontrolador.Necesitamosotra
entradapara señalizarla entradaen modo purga,para lo cual utilizamos PG0y aprovechamosel pull-up
interno. Tomamos la salida de PC5. El circuito se alimenta a 12V, según la especificación del GP8F-R2.

Software

Port serie asincrónico
Utilizamosunade las notas de aplicación deHoltek: HA0004e,y dado que el microcontrolador no tiene otra

cosainteresanteque hacermientrasespera por el ID,realizamosel port serie asincrónico con retardos por
software como indica dicha nota. Incluimos los archivos para facilitar el desarrollo

Manejo de la EEPROM

Utilizamos tambiénunade las notas de aplicación deHoltek: HA0086e,que nos reduce el problema allamar
a una subrutina. Incluimos los archivos para facilitar el desarrollo.
  La primera vez que se ejecute el programa, deberá tener 00 en la primera posición de EEPROM, para detectar
que debecargarla tarjetamaestra.Paraesto,utilizamos el DataEEPROMEditor (Tools-> Data EEPROM
Editor) y hacemos download del archivo acceso.mem que hemos incluido.

Obtención de un ID
Comoviéramos al describir elGP8F-R2,los 40 bits del ID sontransmitidoscomo10 bytesen hex, es decir,

dos bytes ASCII representanun byte (8 bits) del ID, ordenadosnibble alto-nibble bajo. Por consiguiente,
leeremos10 bytes conteniendo 10nibbles y los convertiremosy ordenaremos,como se ve en el listado a
continuación. 

id_vars .section 'DATA'
aux db ? ; auxiliar para cálculos
charcnt db ? ; cuenta de caracteres
IDbuf db 5 DUP(?)

NUMIDs EQU 0 ; # IDs guardadas
IDs EQU 1 ; dirección donde comienzan las IDs

id_code .section 'CODE'

getID: call receive ; Espera STX
sub A,2 ; STX ?
snz Z
jmp getID ; No, loop
mov A,5
mov charcnt,A ; Caracteres a recibir (2x)
mov A,OFFSET IDbuf ; Inicializa puntero
mov MP0,A

rxloop: call receive ; Siguiente caracter
call ASCII2bin ; a binario
mov IAR0,A ; en el buffer
swap IAR0 ; nibble alto
call receive ; Siguiente caracter
call ASCII2bin ; a binario
orm A,IAR0 ; en buffer (nibble bajo)
inc MP0 ; siguiente byte de ID
sdz charcnt ; loop
jmp rxloop
call receive ; espera CR, descarta
call receive ; espera LF, descarta
call receive ; espera ETX
sub A,3 ; ETX ?
snz Z
jmp getID ; No, loop
ret

ASCII2bin: sub A,041h
snz C
add A,7
add A,10
ret

Operaciones con IDs
Deberemosdetectar si el IDrecibidocorrespondea alguno almacenado en labase,borrarla base,y almacenar

un nuevo ID.
Para agregarun ID, por simpleza,dado que nosoportamosborradoselectivode IDs, incrementamosel

indicadordecantidadde IDs almacenadosy agregamos elnuevoID al final de labase.De haberse saturado la
capacidad de la base ignoramos el ID:

addID: call numID ; Obtiene cantidad de IDs en A
mov aux,A ; guarda en aux
sub A,MAXIDs ; Base llena ?
sz C
ret ; ignora
mov A,aux
inc ACC ; incrementa NUMIDs

mov EEDATA,A
call EWEN ; Habilita escritura
call WRITE ; guarda en EEPROM NUMIDs
clr C ; dirección en EEPROM (5xNUMIDs)
mov A,aux
rlc aux ; x2
rlc aux ; x4
add A,aux ; x5
mov EEADDR,A
inc EEADDR ; +1 (NUMIDs)
mov A,5
mov charcnt,A
mov A, OFFSET IDbuf ; Inicializa puntero
mov MP0,A

wIDloop:
mov A,IAR0 ; obtiene del buffer
mov EEDATA,A
call WRITE ; guarda en EEPROM
inc MP0 ; siguiente byte en buffer
inc EEADDR ; siguiente byte en EEPROM
sdz charcnt ; loop
jmp wIDloop
jmp EWDS  ; Inhabilita escritura

Reconocemosun ID almacenadomediantecomparacionessucesivas,buscandoen forma secuencialdesdeel
primero hasta el último:

matchID:
clr EEADDR ; obtiene NUMIDs
call READ  ; lee EEPROM
mov A,EEDATA
mov aux,A  ; inicializa contador
inc EEADDR ; Apunta a principio de base

mIDl2: mov A,5 ; longitud de ID
mov charcnt,A
mov A,OFFSET IDbuf ; Inicializa puntero
mov MP0,A

mIDl1: call READ  ; lee EEPROM
mov A,EEDATA
sub A,IAR0 ; compara con buffer
snz Z ; match ?
jmp mIDnext ; no, siguiente ID
inc MP0 ; sí, byte siguiente en buffer
inc EEADDR ; y en EEPROM
sdz charcnt ; loop
jmp mIDl1
mov A,aux ; MATCH!: ID# en A y Z reset
clr Z
ret

mIDnext:
mov A,charcnt ; apunta a siguiente ID
addm A,EEADDR
sdz aux ; hay suficientes ?
jmp mIDl2
set Z
ret ; No match, devuelve Z set

  Para borrar la base, simplemente hacemos que la cantidad de IDs almacenados sea 1:

purgeDB:
call EWEN  ; Habilita escritura
clr EEADDR ; apunta a NUMIDs
mov A,1 ; borra todo (excepto 1º ID)
mov EEDATA,A
call WRITE ; hecho
jmp EWDS ; Inhabilita escritura

 
Control de Acceso

Ya estamosentonces encondicionesde realizarel control deacceso,el cual implementael protocoloque
propusimos al comienzo de este artículo:

Purge equ PG.0

vars .section 'DATA'
IDs db ? ; # IDs en base

reset .section at 0 'CODE'
ORG 0
jmp start

ORG 00Ch
start:  clr PC

clr PB 
        set PAC ; PA0-7 = inputs

        mov a,0F1h ; PB1-3 = outputs, PB0,4-7 = inputs
mov PBC,a
clr PCC ; PC0-7 = outputs
set PGC ; PG0 = input
mov A,01h
mov BP,A ; bank 1 (EEPROM 'pins')
mov A,040h
mov MP1,A ; MP1 = EECR address
call numID ; Obtiene # IDs en A
mov IDs,A
sz ACC ; Z = ninguno, obtiene master
jmp main
jmp GetMaster

main: call getID ; Espera un ID
call matchID ; La busca en la base
sz Z ; Match ?
jmp main ; No, loop
sub A,IDs ; Mira si es el Master
sz Z ; Master ?
jmp Special ; Sí, modo Autoriza/purga

open: call OpenLock ; No, abre
jmp main

Special:
call pulseLock ; Acknowledge (2 pulsos cortos)
call HSdelay
call pulseLock
sz Purge ; purga ?
jmp DOPurge
call getID ; no, espera un ID
call matchID ; lo busca en base
sz Z ; Match ?
jmp learn ; No, lo aprende
sub A,IDs ; Sí, se fija si es el Master
snz Z ; Master ?
jmp Special ; No, ignora
jmp open ; Sí, modo normal (abre)

learn: call addID ; agrega a la base
call OpenLock ; abre (ignora errores)
jmp start

DOPurge:
call purgeDB ; Purga base
call HSdelay ; Acknowledge (3 pulsos cortos)
call pulseLock
jmp start ; restart

GetMaster:
call pulseLock ; Acknowledge (1 pulso corto)
call HSdelay
call getID ; Espera un ID
call addID ; lo guarda
call pulseLock ; Acknowledge (1 pulso corto)
call HSdelay
jmp start ; restart

Lasrutinasdesarrolladasfueron escritas teniendo en cuenta lasespecificacionesdel fabricante,para4MHz de
clock. La cantidadde callsanidadasse comprobóparaun HT48E30.Paraotras aplicaciones, se sugiere al
interesado revisar que se cumpla lacantidadde calls permitidos por elhardwarestack del micro deHoltek
utilizado.

